

# The use of the algebraic programming in teaching general relativity \*

Florin A. Ghergu and Dumitru N. Vulcanov

*Theoretical and Computational Physics Department*

*The West University of Timișoara*

*Bv. V. Pârvan No. 4, 1900 Timișoara, România*

*E-mail : florin/vulcan@mitica.physics.uvt.ro*

## Abstract

The article presents some aspects on the use of computer in teaching general relativity for undergraduate students with some experience in computer manipulation. The article presents some simple algebraic programming (in REDUCE+EXCALC package) procedures for obtaining and the study of some exact solutions of the Einstein equations in order to convince a dedicated student in general relativity about the utility of a computer algebra system.

## 1 Introduction

Teaching general relativity is a very difficult task not only for the "teacher" but also for the students. But why ?

General relativity (GR) is not only a theory of gravity; it is a theory of the structure of space and time, and hence a theory of the dynamics of the universe in its entirety. Thus, the theory is a vast edifice of **pure geometry**, indisputably elegant, but of a great **mathematical difficulty** especially for undergraduate students (among others...). But why undergraduate students? Because, in our department, (considering GR a necessity for the medium

---

\*accepted for publication in Computing in Science and Engineering, 2000

graduate physics people) we make efforts to introduce a course of GR at the III-rd year level. After some years of experience in teaching GR at graduate students, specialized in theoretical physics, we can say that after some weeks of introducing the most important tools of differential geometry, starting with the physical problems of GR, an important part of our students were almost discouraged. At an undergraduate level, with unspecialized students, this risk is much greater.

Thus, algebraic programming systems (like REDUCE - [1]) which contain differential geometry packages can become a very important tool for surpassing these difficulties. With the computer, the student can learn very fast, and in an attractive manner, the important notions of differential geometry, tensor calculus and, of course, the exterior calculus (with EXCALC for REDUCE, for example - [2]). As an example, he can make, after some simple computer manipulation, the long and unattractive (for the "manual" student) calculations from the Riemannian geometry, with metrics and Christoffel symbols, etc. Thus, the first section of this article illustrates how we can use the EXCALC package in teaching Riemannian geometry.

But having such a powerful tool for calculus in differential geometry, we can try to introduce some aspects of GR using computer algebra systems. For example, the Schwarzschild solution can be easily obtained on the computer, and also we can generate the Reissner-Nordstrom solution as exact solutions of the Einstein equations. These aspects are presented in the second section of the article.

The last section of the article presents the use of computer algebra in finding and treating other exact homogeneous solutions of the Einstein equations with cosmological constant (de Sitter and anti de Sitter metrics more precisely).

As a conclusion, we consider the use of computer as an important tool for teaching general relativity. During the last two years we have experienced several packages of procedures, (in REDUCE + EXCALC for algebraic programming and in Mathematica or Maple for graphic visualizations) which fulfill this purpose. Even when the students were real beginners in computer manipulation we have obtained visible good results, in approaching several topics of differential geometry and of course, in general relativity.

## 2 Differential geometry in EXCALC

The program EXCALC ([2]) is completely embedded in REDUCE, thus all features and facilities of REDUCE are available in a calculation.

EXCALC is designed for easy use by all who are familiar (or want to become) with the calculus of Modern Differential Geometry. The program is currently able to handle scalar-valued exterior forms, vectors and operations between them, as well as non-scalar valued forms (indexed forms).

Geometrical objects like **exterior forms** or **vectors** are introduced to the system by declaration commands; therefore zero-forms (functions) must also be declared. Also, specific operations with geometric objects are available in EXCALC like : **exterior multiplication** between exterior forms (carried out with the nary infix operator  $\wedge$  (wedge)), **partial differentiation** (is denoted by the operator @), **exterior differentiation** of exterior forms (carried out by the operator d), the **inner product** between a vector and an exterior form (represented by the diphthong  $\_ |$  (underscore or-bar)), the **Lie derivative** can be taken between a vector and an exterior form or between two vectors (represented by the infix operator  $| \_$ ), the **Hodge-\* duality** operator (maps an exterior form of degree K to an exterior form of degree N-K, where N is the dimension of the space). It is possible to declare an indexed quantity completely antisymmetric or completely symmetric. Some examples :

```

PFORM U=1,V=1,W=K;    %declaration of some forms
(3*U-A*W)^(W+5*V)^U;
A*(5*U^V^W - U^W^W)
@(SIN X,X);          % partial differentiation
COS(X)

PFORM X=0,Y=K,Z=M;
D(X * Y);             % exterior differentiation of a
X*d Y + d X^Y         % product of two forms
D(X*Y^Z);
      K
( - 1) *X*Y^d Z  + X*d Y^Z + d X^Y^Z

PFORM X=0,Y=K,Z=M;   TVECTOR U,V;
```

```

U_|(X*Y^Z);          % inner product
      K
X*(( - 1) *Y^U_|Z + U_|Y^Z)

PFORM Z=K; TVECTOR U;
U |_| Z;              % Lie derivative
U_|d Z + d(U_|Z)

```

A **metric structure** is defined in EXCALC by specifying a set of basis one- forms (the coframe) together with the metric. The clause WITH METRIC can be omitted if the metric is Euclidean and the shorthand WITH SIGNATURE <diagonal elements> can be used in the case of a pseudo-Euclidean metric. The splitting of a metric structure in its metric tensor coefficients and basis one-forms is completely arbitrary including the extremes of an orthonormal frame and a coordinate frame. Examples ([2]) :

```

COFRAME O(T)=D T, O X=D X
WITH SIGNATURE -1,1;      %A Lorentz coframe;

COFRAME E R=D R, E PH=D PH          %Polar coordinate
WITH METRIC G=E R*E R+R**2*E PH*E PH; %basis;

```

The frame, dual to the coframe defined by the COFRAME command can be introduced by FRAME <identifier>. This command causes the dual property to be recognized, and the tangent vectors of the coordinate functions are replaced by the frame basis vectors.

The command RIEMANNCONX is provided for calculating the connection 1 forms. Example : calculate the connection 1-form and curvature 2-form on S(2) (displaying only the nonzero results) :

```

COFRAME E TH=R*D TH,E PH=R*SIN(TH)*D PH;
RIEMANNCONX OM;
OM(K,-L);          %Display the connection forms;

      PH      PH
NS      := (E *COS(TH))/(SIN(TH)*R)
      TH

```

```

      TH      PH
NS      := ( - E *COS(TH))/(SIN(TH)*R)
      PH

PFORM CURV(K,L)=2;
CURV(K,-L):=D OM(K,-L) + OM(K,-M)^OM(M-L);
                                     %The curvature forms

      PH      TH PH 2
CURV      := ( - E ^E )/R %it was a sphere with
      TH                                     %radius R.

      TH      TH PH 2
CURV      := (E ^E )/R
      PH

```

### 3 General relativity on the computer. Schwarzschild solution

The students in our Faculty of Physics are, generally speaking, well trained in practical computer manipulations. There is no semester without at least one course with labs in the computer room. But when we invited our students to come in the computer room, to learn something about general relativity with the computer, it was a general surprise, because they considered (until now) the computer as a tool for hard **numerical** computations. They do not know almost anything about **computer algebra** - [3].

It is not necessary to use sophisticated procedures, with large and complicated metric statements (which are almost impossible to calculate by hand in a civilized time of teaching) in order to convince a dedicated student in general relativity about the utility of a computer algebra system. It is enough to run a simple program like ([4]) :

```

pform psi=0; fdomain psi=psi(r);
coframe
  o(t)      = psi          * d t,    % Schwarzschild
  o(r)      = (1/psi)      * d r,    % metric

```

```

o(theta) = r * d theta,
o(phi)   = r * sin(theta)* d phi
with signature 1,-1,-1,-1; frame e;

```

to introduce a Schwarzschild type metric in spherical coordinates  $(r, \theta, \varphi)$ . This means that in classical notation the interval is

$$ds^2 = \Psi^2 dt^2 - \frac{1}{\Psi^2} dr^2 - r^2 (d\theta^2 + \sin^2 \theta d\varphi^2) \quad (1)$$

$\Psi$  being a function  $\Psi = \Psi(r)$  and differ from a Minkowski one by the a new "unknown" function ( $\Psi = \sqrt{1 + \text{unknown}(r)}$ ) which must be determined from Einstein equations :

```

pform unknown=0; fdomain unknown=unknown(r);
psi := sqrt(1 + unknown);

```

Now comes the most important part of the procedure : the calculation of the components of Einstein tensor (*einstein3*) via the Riemann or Levi-Civita connection 1-form  $\Gamma^{ij}$  - *christ1* and the curvature 2-form  $R^{ij}$  - *curv2*

```

pform chris1(a,b)=1, curv2(a,b)=2, einstein3(a)=3;
antisymmetric chris1, curv2;
riemannconx christ1; chris1(a,b) := christ1(b,a);
curv2(a,b) := d chris1(a,b) + chris1(-c,b) ^ chris1(a,c);
einstein3(-a) := (1/2) * curv2(b,c) ^ #(o(-a)^o(-b)^o(-c));

```

The last of the above program lines just defines the Einstein 3-form which appears in the Einstein equations. Those who prefer the coordinate components form (thus the Einstein tensor  $G_{ij} = R_{ij} - \frac{1}{2}g_{ij}R$ ) can use the next line to "pick-up" these components as :

```

pform Ein(i,j)=0;
Ein(-i,-j):=e(-i)_|#einstein3(-j);

```

A typical component ( $G^\phi$ ) of the output of *einstein3* reads :

$$- \begin{matrix} \text{T} & \text{R} & \text{THETA} \\ (0 & \wedge & \wedge \end{matrix} \begin{matrix} *(@ & \text{UNKNOWN}*R & + & 2*(@ & \text{UNKNOWN})) \\ & \text{R} & \text{R} & & \text{R} \end{matrix} / (2*R)$$

or :

$$G^\phi = \left( -\frac{1}{2} \frac{\partial^2 unknown}{\partial r^2} - \frac{1}{r} \frac{\partial unknown}{\partial r} \right) o^t \wedge o^r \wedge o^\theta$$

Requiring the coefficients to vanish yields a second order differential equation for the function *unknown*. Trying  $unknown = \alpha * m/r * n$ , after using SOLVE ([1]) package (i.e. the EXCALC command **solve(einstein3(-phi),unknown);** ), we obtain  $n = -1$  and :

`unknown := - alpha * m/r;`

where "m" is the mass and *alpha* a constant coefficient to be determined by physical considerations (link to the Newtonian theory, for example).

Finally, evaluating the  $\psi$  function ( $psi := psi$ ), we obtain :

$$\psi = \sqrt{1 - \frac{\alpha m}{r}}$$

or, in (1) we have

$$ds^2 = \left( 1 - \frac{\alpha m}{r} \right) dt^2 - \frac{1}{1 - \frac{\alpha m}{r}} dr^2 - r^2 (d\theta^2 + \sin^2 \theta d\varphi^2) \quad (2)$$

which is the typical form of the Schwarzschild metric ([4]-[5]) identifying  $\alpha = 2$  by physical considerations.

From now on, it is possible to study, in a similar way more complex situations, like Reissner-Nordström metric (starting, of course with the above Schwarzschild one - this example is presented in detail in [6]). The teacher can select more exact solutions of Einstein equations in order to complete the education of his students. Also, algebraic programming can be used to present (in a very fast manner) the canonical version of general relativity ([7]) or the post-Newtonian approximation ([4]).

The next section is dedicated to more examples which can be used in the teaching process of general relativity.

## 4 Other examples

According to the literature (see, for example [4] or [8]) two of the homogeneous solutions of the Einstein equations with cosmological constant are the

de Sitter and anti-de Sitter metrics, having the line element written as :

$$ds^2 = e^{ng} dt^2 - e^{kf} dx^2 - e^{ng+kf} (dy^2 + dz^2) \quad (3)$$

where  $f$  and  $g$  are two unknown functions of  $x$  and  $t$  variables which we shall determine by imposing that the vacuum Einstein equations with cosmological constant to be fulfilled :

$$G_{ij} = R_{ij} - \frac{1}{2} g_{ij} R - \Lambda g_{ij} = 0 \quad (4)$$

Also in the above equation (3)  $n$  and  $k$  are two constants introduced in order to control the two solutions : for  $n = 0$  and  $k = 1$  we shall obtain the de Sitter solution and for  $n = 1$  and  $k = 1$  we have the anti-de Sitter solution. Thus we can use a program sequence for introducing this metric as :

```
pform f=0,g=0; fdomain f=f(x,t),g=g(x,t);
coframe
  o(t)      =  exp(n*g/2)          * d t,
  o(r)      =  exp(k*f/2)          * d x,
  o(theta)  =  exp((n*g+k*f)/2)    * d y,
  o(phi)    =  exp((n*g+k*f)/2)    * d z
with signature 1,-1,-1,-1; frame e;
```

Now we must run the sequences which calculates the Einstein tensor  $G_{ij}$  components with cosmological, namely :

```
riemannconx chris;
pform chris1(a,b)=1;
pform curv2(a,b)=2;
pform einstein3(a)=3;
antisymmetric chris1;
antisymmetric curv2;
chris1(a,b):=chris(b,a);
curv2(a,b):= d chris1(a,b)+chris1(-c,b)^chris1(a,c);
einstein3(-a):=(1/2)*curv2(b,c)^ #o(-b)^o(-c)^o(-a))-Lam*#o(-a);
pform Ein(i,j)=0;
Ein(-i,-j):=e(-i)_l(#einstein3(-j));
```



where **Lam** represents the cosmologic constant  $\Lambda$  and **Ein(-i,-j)** represents the components of the Einstein tensor  $G_{ij}$  with cosmological constant in coordinate frame. A typical output is, for example :

$$\begin{aligned}
\text{Ein}_{T T} := & \left( -4 * E_{T T} \frac{F * K + G * N}{T} + 3 * E_{T T} \frac{F * K}{T} \right. \\
& + 4 * E_{T T} \frac{F * K}{T} * @ F * @ G * K * N + E_{T T} \frac{F * K}{T} * @ G * N \\
& - 4 * E_{T T} \frac{G * N}{X X} * @ F * K - E_{T T} \frac{G * N}{X} * @ F * K \\
& - 4 * E_{T T} \frac{G * N}{X} * @ F * @ G * K * N - 4 * E_{T T} \frac{G * N}{X X} * @ G * N \\
& \left. - 3 * E_{T T} \frac{G * N}{X} \right) / (4 * E_{T T} \frac{F * K + G * N}{T})
\end{aligned}$$

which means, in “normal” form :

$$\begin{aligned}
G_{tt} = e^{-ng} & \left[ \frac{3}{4} k^2 \left( \frac{\partial f}{\partial t} \right)^2 + kn \frac{\partial f}{\partial t} \frac{\partial g}{\partial t} + \frac{1}{4} n^2 \left( \frac{\partial g}{\partial t} \right)^2 \right] - \\
& e^{-kf} \left[ k \frac{\partial^2 f}{\partial x^2} + \frac{1}{4} k^2 \left( \frac{\partial f}{\partial x} \right)^2 + kn \frac{\partial f}{\partial x} \frac{\partial g}{\partial x} + n \frac{\partial^2 g}{\partial x^2} + \frac{3}{4} n^2 \left( \frac{\partial g}{\partial x} \right)^2 \right] - \Lambda
\end{aligned}$$

Now assigning to the parameters the values  $n = 0$  and  $k = 1$  the above component of the Einstein tensor) and the corresponding Einstein equation) will be :

$$G_{tt} = \frac{3}{4} \left( \frac{\partial f}{\partial t} \right)^2 - e^{-f} \left[ \frac{\partial^2 f}{\partial x^2} + \frac{1}{4} \left( \frac{\partial f}{\partial x} \right)^2 \right] - \Lambda = 0$$

By inspecting this equation ([9]) we can observe that his general solution can be written as  $f(x, t) = f1(t) + f2(x)$  where

$$\left(\frac{df1(t)}{dt}\right)^2 = \frac{1}{3}C_1 e^{-f1(t)} + \frac{4}{3}\Lambda \quad \text{and}$$

$$\frac{d^2 f2(x)}{dx^2} = \frac{1}{4}C_1 e^{f2(x)} - \frac{1}{4}\left(\frac{df2(x)}{dx}\right)^2$$

where  $C_1$  is a constant. Thus we can choose  $f(x, t) = f1(t)$  such that the above equation becomes :

```
pform f1=0;fdomain f1=f1(t);f:=f1;
Ein(-t,-t);
      2
3*@ F1  - 4*LAM
      T
-----
      4
```

as computer output and

$$G_{tt} = \frac{3}{4}\left(\frac{df1}{dt}\right)^2 - \Lambda = 0$$

This equation has an obvious solution of the form :

$$f1 = f = \frac{2}{3}\sqrt{3\Lambda}t + C_1$$

which fulfill the Einstein equations with cosmological constant (all the components os **Ein(-i,-j)** 0-form are zero). Then the line element becomes :

$$ds^2 = dt^2 - e^{\frac{2}{3}\sqrt{3\Lambda}t+C_1} (dx^2 + dy^2 + dz^2)$$

which is, of course the well-known De Sitter solution. For the anti-De Sitter solution we must choose the parameters as  $n = 1$  and  $k = 0$ . Thus the interesting Einstein tensor components become :

n:=1;k:=0;  
Ein(-t,-t);

$$\begin{array}{c}
\begin{array}{ccccccc}
& G & & G & 2 & & G \\
- & 4 * E & * @ & G & - & 3 * E & * @ & G & - & 4 * E & * LAM & + & @ & G \\
& & X & X & & & X & & & & & & T & \\
\end{array} \\
\hline
\begin{array}{c}
G \\
4 * E
\end{array}
\end{array}$$

Ein(-x,-x);

$$\begin{array}{c}
\begin{array}{ccccccc}
& G & 2 & & G & & 2 \\
3 * E & * @ & G & + & 4 * E & * LAM & - & 4 * @ & G & - & @ & G \\
& & X & & & & & T & T & & T & \\
\end{array} \\
\hline
\begin{array}{c}
G \\
4 * E
\end{array}
\end{array}$$

as computer output, or in “normal” transcript :

$$\begin{aligned}
G_{tt} &= \frac{1}{4} e^{-g} \left( \frac{\partial g}{\partial t} \right)^2 - \frac{\partial^2 g}{\partial x^2} - \frac{3}{4} \left( \frac{\partial g}{\partial x} \right)^2 - \Lambda = 0 \\
G_{xx} &= \frac{3}{4} \left( \frac{\partial g}{\partial x} \right)^2 - e^{-g} \left[ \frac{\partial^2 g}{\partial t^2} + \frac{1}{4} \left( \frac{\partial g}{\partial t} \right)^2 \right] + \Lambda = 0
\end{aligned}$$

Solving, in a similar way as in the De Sitter case, the second of the above Einstein equations, we have to impose  $g(x, t) := g_1(x)$ . Thus we obtain a solution as :

$$g = g_1 = \frac{2}{3} \sqrt{-3\Lambda} x + C_2$$

where  $C_2$  is a new constant. We have the line element as

$$ds^2 = -dx^2 + e^{\frac{2}{3}\sqrt{-3\Lambda}x+C_1} (dt^2 - dy^2 - dz^2)$$

which is the anti-De Sitter solution.

The REDUCE+EXCALC procedures we presented here are also suitable for a series of other applications during the relativity lab classroom (...). For example the teacher can propose to the students to check if other metrics fulfill the Einstein equations. They can use the above program sequences with minor modifications. As an exercise we propose here the Ozsvath metric ([8])

$$ds^2 = dx^2 + e^{2\sqrt{-\Lambda/3}x} (dy^2 + 2dudv) + fe^{-\sqrt{-\Lambda/3}x} (-2\sqrt{2}dy + fe^{-3\sqrt{-\Lambda/3}x}dv) dv$$

which is also an homogeneous solution of vacuum Einstein equations with cosmological term. Here a special care must be taken to the changing of the signature of the metric and to the nondiagonal terms of the metric.

## References

- [1] A.C. Hearn, REDUCE User's Manual, Version 3.3, RAND Publications, The RAND Corp., Santa Monica, CA, 90407-2138, 1987
- [2] E. Schrüfer, EXCALC : A system for doing Calculus of Modern Differential Geometry, GMD Institut II, D-53757, St. Augustin, Germany, 1994
- [3] D. Stauffer, F.W. Hehl, N. Ito, V. Winkelmann, J. G. Zabolitzky, Computer Simulation and Computer Algebra, Springer Verlag, Berlin, 1993
- [4] C.W.Misner, K.S.Thorne, J.A.Wheeler : Gravitation, W.H.Freeman Co., San Francisco, 1973
- [5] S.Hawking, G.F.R.Ellis : The large scale structure of space-time, Cambridge Univ. Press, Cambridge, 1973
- [6] R. Puntigam, E. Schrüfer, F.W. Hehl, The use of computer algebra in Maxwell's theory, Computer Algebra and Engineering, Proc. of the XiF-Workshop, Bielefeld, Aug. 1994, J Fleisher et al. (eds.), World Scientific, Singapore (1995)
- [7] D.N. Vulcanov, Algebraic Programming in the Hamiltonian Version of General Relativity, in Relativity and Scientific Computing, F.W. Hehl, R.A. Puntigam and H. Ruder (eds.), Springer Verlag, 1996

- [8] I. Ozsvath, “All homogeneous solutions of Einstein’s vacuum field equations”, in W. Rindler and A. Trautman (eds), *Gravitation and Geometry*, Bibliopolis, Napoli, 1987
- [9] M.E. Taylor, *Partial Differential Equations*, Springer Verlag, 1996